Tran Duong HIEN
WESTERN POMERANIA UNIVERSITY OF TECHNOLOGY,
DEPARTMENT OF STRUCTURE THEORY

A computer code for eliminating secularity in perturbed systems

Prof. dr hab. inż. Tran Duong HIEN

Specjalność - matematyka stosowana, mechanika komputerowa, informatyka. Od 1978 do 1995 - pracownik Instytutu Podstawowych Problemów Techniki Polskiej Akademii Nauk; 1990 - habilitacja w IPPT PAN. Od 1995 - Profesor Politechniki Szczecińskiej. Autor i współautor 5 książek, wśród których 2 wydane przez Wiley'a, ok.150 prac opublikowanych w kraju i zagranicą, m.in.w Int.J. Numer. Meth. Eng, Int.J. Comput. Meth. Appl. Mech. Eng, Int.J. Comput. Struct., Int.J. Eng. Comput.

e-mail: tdhien@ps.pl

Abstract

In the paper an efficient algorithm is formulated to eliminate inherent secular effects in solving engineering problems by the perturbation technology. Basing upon the modified data-window concept, a computer procedure written in Fortran is worked out in the standard form and can be applied to deal with a large class of nonlinear and/or stochastic engineering systems of dynamics, control, identification and optimization. An illustrating numerical example related to a structural dynamics system with random parameters is presented.

Keywords: secularity, perturbation, sensitivity, nonlinearity, stochastics.

Procedura komputerowa usunięcia sekularności w układach perturbowanych

Streszczenie

W pracy sformułowano skuteczny algorytm wyeliminowania efektów sekularnych, swoistych w rozwiązywaniu zagadnień inżynierskich drogą perturbacyjną. Bazując na zmodyfikowanym pojęciu okien-danych, opracowano w języku Fortran podprogram komputerowy. Będący w standardowej formie, kod ten może być stosowany w szerokiej klasie nieliniowych lub/i stochastycznych układów dynamiki, kontroli, identyfikacji czy optymalizacji. Przedstawiono ilustrujący przykład dotyczący dynamiki konstrukcyjnego układu z parametrami losowymi.

Słowa kluczowe: sekularność, perturbacja, wrażliwość, nieliniowość, stochastyka.

1. Introduction

It is well-known that the perturbation methodology takes a crucial role in solving many time-varying problems in engineering, for nonlinear and stochastic systems in particular, cf. [1, 2, 3] for instance. The main idea behind the approach is based on the exponential expansion of the system response, defined as a state variable vector, around a critical point, adopted as a system parameter vector. By equating the same-order derivative terms in the expanded expressions a set of hierarchical equations is obtained.

The significant advantage of the governing hierarchical equation system is that the same linear differential operator acts on the left-hand side, with the whole system nonlinearity and/or randomness being translated into the right-hand side vectors. An inherent drawback of the perturbation technology, however, is that since the eigensystem terms, for example system natural frequencies in dynamics, are involved in the right-hand side vectors of higher-order equations. That is why the unavoidable fact is that the perturbational solutions, increasing step-by-step with respect to time, tend to infinity. Such a resonance-type effect is known as secularity. A numerical algorithm for eliminating the secular terms, necessarily needed in any computationally perturbational procedure, is thus the goal of the paper.

2. Secularity elimination

In the second-order perturbation method we accept that the expansion is limited to the first two terms. This can produce invalid solutions due to the appearance of the secular terms, which grow indefinitely with time. Such unbounded solutions may occur even for systems that are known to possess bounded solutions, such as conservative systems. This effect is much more significant in the stochastic and sensitivity analysis, since the expectations of the sensitivity gradient coefficients are bilinear functions of the unknowns of the system at hand, while their covariances are square functions of these quantities. It is shown above that the natural frequencies or linear operators at the left-hand sides are the same for all the zeroth-, first- and second-order equations in the systems. This is also the case for the transformed equations and the uncoupled systems, described on the normal-coordinate modal base. The zeroth-order equations are only excited by zeroth-order external forces so generally no resonance occurs. However, the first- and second-order forcing sequences are functions of the zeroth-order response that ensure resonant excitation until the transient part of the zeroth-order solution is damped away. Hence, a rough application of the second-order perturbation approach whereby only the amplitude is altered may not always be satisfactory; and the resonant part involved on the right-hand side of the first- and second-order equations has to be necessarily removed or weighted to maintain validity of the solution when a relatively long duration of the system response is required. In other words, a modification to prevent the occurrence of the secular terms must alter both the amplitude and the period of vibrations.

To deal with the secularity problem some theoretical methodologies have been developed in [1, 4, 5, 6]. However, most of this work was concerned with single-degree-of-freedom systems treated analytically; little work has been done on the numerical elimination of secularities for multi-degree-of-freedom systems. An efficient algorithm based on the fast Fourier transform of the sine and cosine pair was developed in [7]. The numerical procedure presented below is an alternative form of it. Namely, the real-valued signals of the forcing functions is discretized and then converted into a complex-valued sequence of half length, the fast Fourier transform being carried out on the complex-valued sequence. The concept of the discrete Fourier transform and the philosophy of the fast Fourier transform (FFT) and distinguishing advantages of this modern computational technique have been described in detail in [8]. We shall briefly discuss here how to use FFT to eliminate secularities in practical engineering computation.

Let us consider a time-varying forcing function $Q(\tau)$, with the time variable $\tau \in [0,T]$, treated as an input signal of a discrete system, i.e. as a sequence of sample values $Q(t_n)$ in which $t_n = n\Delta t$, n being an integer number while Δt is the sampling interval. Employing the data-window concept of the zero-order hold we generate $Q(\tau)$ in the form of a staircase, a sampling interval apart, as

$$Q_n = Q(t_n) = (n\Delta t), \tag{1}$$

where $t_n \in [n\Delta t, (n+1)\Delta t]$, $n = 1,2,\dots,2L$, with $L = T/\Delta t$ being the sample length. Next, the discrete real-valued signal is assembled and converted into a complex-valued sequence of the discrete time signal p_k of the form

$$p_k = p(t_k) = Q(t_{2k}) + iQ(t_{2k+1}), \tag{2}$$

where the symbol ‘i’ denotes the imaginary unit and L now becomes the half sample length, being the Fourier transform degrees or the complex sequence length.

In what follows, we shall use the angular velocity, i.e. the circular frequency $\omega=2\pi f$, rather than the cyclic frequency f used for the derivations in [8]. This is because with the symmetry of the factors outside the integrals the f-representation is more elegant in the mathematical sense, whereas in computational applications the ω-representation directly provides a physical interpretation for the solution of the system eigenproblem. In view of this the finite-range direct Fourier transform $P(\omega,T)$ of p_k can be discretized via a sequence of discrete signals in the form, cf. [8]

$$P_j(\omega,T)=P(\omega_j,T)=\frac{T}{L}\sum_{k=0}^{L-1}p(t_k)\exp(-i\omega_j t_k)=\frac{T}{L}\sum_{k=0}^{L-1}p_k\exp\left(-i\frac{2\pi}{L}\right)jk, \tag{3}$$

where $k = 1,2,...,L\text{-}1,\ \omega_j = j\Delta\omega;\ \Delta\omega = 2\pi/T;\ \omega_j\in[0,\Omega];\ \Omega=\pi/\Delta t.$ The frequency range Ω is known as the circular Nyquist frequency, being the highest frequency appearing in the data record corresponding to the sampling interval Δt. The choice of an appropriate value for the sampling interval and, if necessary, of a data processing technique for saving the computational cost and simultaneously handling the aliasing problem, is discussed in [3, 8]. The algorithm for the fast Fourier analysis on a complex-valued sequence, cf. [8], can be used to evaluate the discretized one-sided Fourier spectrum $P(\omega,T)$ of the time sequence of the input signal $p_k(\tau,T)$ in the form of a discretized frequency sequence.

It is observed from the higher-order hierarchical equations that since the first- and second-order modal forcing records are already uncoupled, only the terms related to their corresponding natural frequencies will produce secularity. To remove the secular terms the coefficients P_j(ω)of the frequency sequence which lie within a specified range of the modal systems are eliminated or weighted adequately, thereby removing the resonant part from Fourier spectra. In other words, for the N-degree-of-freedom system considered the coefficients associated with the j-th frequencies adjacent to a fixed n-th natural frequency ω_n, n=1,2,…,N, are almost entirely weighted (eliminated), whereas the coefficients separated from this frequency domain are unaffected. Thus, in the direct Fourier transform the coefficients of a frequency sequence are assumed to cause secularity when they satisfy the condition

$$\omega_n-\Delta\Omega\le\omega_j\le\omega_n+\Delta\Omega, \tag{4}$$

where $\Delta\Omega$ is a prescribed frequency range. There exist many data windows to filtrate secular frequencies; some weighting functions frequently applied are given as follows, cf. [5, 6]:

— triangular

$$P_j := P_j\frac{\left|\omega_{(n)}-\omega_j\right|}{\Delta\Omega}, \tag{5}$$

— cosine

$$P_j := P_j\left[1-\cos\frac{\pi\left(\omega_{(n)}-\omega_j\right)}{2\Delta\Omega}\right], \tag{6}$$

— cosine-squared

$$P_j := \frac{1}{2}P_j\left[1-\cos\frac{\pi\left(\omega_{(n)}-\omega_j\right)}{\Delta\Omega}\right]. \tag{7}$$

Once the coefficients P_j of the frequency sequence are weighted, i.e. the secular terms are eliminated, the discrete inverse Fourier transform of the form

$$p_k(\tau,\Omega)=P(t_k,\Omega)=\frac{1}{T}\sum_{j=0}^{L-1}P(\omega_j)\exp(i\omega_j t_k)=\frac{1}{T}\sum_{j=0}^{L-1}P_j\exp\left(i\frac{2\pi}{L}\right)jk, \tag{8}$$

with k=1,2,…,L-1,, is carried out by the fast Fourier synthesis to obtain the complex-valued forcing record p_k with no resonant part included. Finally, the real-valued forcing sequence in which the secular terms are already eliminated can be recovered from the real and imagine parts of the discrete output signal p_k as

$$Q_{2k}(\tau)=\operatorname{Re}p_k(\tau),\qquad Q_{2k+1}(\tau)=\operatorname{Im}p_k(\tau), \tag{9}$$

with k=1,2,…,L-1. A drawback of the stochastic finite element applications in structural dynamics is that the accuracy may deteriorate even with slight structural damping. In this context the secularity elimination cannot be neglected to ensure that all statistical results are valid for a long response duration. On the other hand, to save the computational cost there is a need to incorporate an effective algorithm to simulate the forcing function with large length in any stochastic finite element code. The use of FFT is most suitable in this regard.

3. Standard routine in Fortran

The routine **`removesecular`** is written in the Fortran fixed format to weight secular terms involved on the right-hand side of the first- and second-order equations of initial and/or initial-terminal problems. To filtrate the secular terms from a discrete time sequence describing the right-hand side of a linear differential equation the following steps have to be accomplished: (i) the discrete finite-range Fourier analysis is carried out for the time sequence to obtain its discrete Fourier spectra, (ii) in the resulting frequency sequence the coefficients at frequencies within the domain extending from $\omega_n-\Delta\Omega$, to $\omega_n+\Delta\Omega$, $\Delta\Omega$ being a prescribed frequency range, are weighted by using a data window, (iii) the discrete finite-range Fourier synthesis is performed for the weighted frequency sequence to recover the time sequence with no resonant part. The frequency range $\Delta\Omega$ is specified by the user through a frequency range factor r (identified by the formal variable **`range`** entered on the control data line for a time-varying response, and defined with respect to the smallest natural frequency ω_1, i.e. $\Delta\Omega=r\omega_1$, while the data window employed is of the cosine-squared type, cf. Eq. (7).

The standard code **`RemoveSecular`** is worked out for dynamic analysis of multi-degree-of-freedom systems with random parameters by using the mode superposition scheme; the FFT routine **`FastFourierTransform`** called in this procedure is presented in [3]. Clearly, the code can be modified to fit into other time-varying systems with no difficulty.

```
      subroutine RemoveSecular (a,accel,displ,freq,sample,
     *                          randdt,nf,nt,ntp1,mtot,n5,iflag,i12)
*      To eliminate secular terms by fast Fourier transform

*      ON INPUT:
*        iflag        Flag for controlling dynamic response process
*        istyp2       Random variable is Young modulus
*        istyp3       Random variable is mass density
*        istyp14      Random variable is cross-sectional area or length
*        i12          Control parameter for response do-loops
*                     1 - primary response; 2 - primary-adjoint response
*        nf           Number of requested normalized coordinates
*        nt           Sampling length (number of time intervals)
*        ntp1         Extended sampling length
*        n5           Tail address of working array a(.)
*        randdt       Sampling time interval
*        range        Frequency range factor for secularity elimination
*        sample(.)    Input time real-valued sequence
*        freq(.,.)    Natural frequencies
```

```
*      ON OUTPUT:
*        accel(.,.)   Weighted acceleration-type time sequence
*        displ(.,.)   Weighted displacement-type time sequence
*        a(.)         Working (mother) array
*      (sample of length nt is extended with sample(nt+1)=sample(nt))

       implicit       real*8 (a-h,o-z)
       logical        istyp14,istyp2,istyp3,iflag0
       common /istyp/ istyp14,istyp2,istyp3
       common /fftdt/ inul,range,i001(5)
       dimension      a(1),accel(nf,nt),displ(nf,nt),freq(nf),sample(ntp1)
       data           pi/3.14159265358979d0/,one/0.999999999999999d0/

       iflag0=iflag.eq.0
       ntm1=nt-1                   ! Check if sampling length is satisfied
       if(iflag0) then                     ! (nt-1.ge.8 and divisible by 4)
         if(ntm1.lt.8.or.ntm1.ne.(ntm1/4*4)) then
           write (6,2001) ntm1
           STOP 'E R R O R !  Bad time-interval number'
         endif
       endif

       nyquist=ntm1/2
       mtwo=2
       do i=1,32                      ! Max. sampling length 2**32 allowed
         if(nyquist.le.mtwo) goto 15
         mtwo=mtwo+mtwo
       enddo
    15 rewind 7
       read  (7) freq                          ! Read in natural frequencies
       domega=range*freq(1)                        ! Secular frequency range
       pidr=pi/domega
       fnyquist=pi/randdt                      ! Set Nyquist cutoff frequency
       deltaf=fnyquist/dfloat(nyquist)         ! Sampling frequency interval

       if(iflag0) then                               ! Set do-loop counters
         if(istyp3) then                 ! Random variable as mass density
           jstart=1
           jend=1
         elseif(istyp2) then            ! Random variable as Young modulus
           jstart=2
           jend=2
         else                ! Random variable as cross-section and length
           jstart=1
           jend=2
         endif
       else
         jstart=3
         jend=3
       endif
       rewind 9
       rewind 10

*      Process primary (mm=1) and adjoint (mm=2) loop over nf frequencies

       do 200 mm=1,i12                               ! Primary-adjoint loop
         do 100 n=1,nf                         ! Normalized coordinate loop
           freqn=freq(n)                          ! n-th natural frequency
           ffirst=freqn-domega                       ! Left-bound frequency
           if(ffirst.le.0.0d0) ffirst=0.0d0
           flast=freqn+domega                       ! Right-bound frequency
           if(flast.gt.fnyquist) flast=fnyquist
           k=(ffirst+one*deltaf)/deltaf                 ! Frequency pointer
           fkm1=deltaf*dfloat(k-1)

         do 100 j=jstart,jend
           read (10) (sample(i),i=1,ntm1)           ! Read in time sequence
           sample(nt)=0.0d0
           sample(ntp1)=0.0d0
                             ! Calculating Fourier spectra by direct FFT
*          call FastFourierTransform (sample,a(n5),ntm1,-1)
                        ! Removing secular terms from frequency sequence
           fk=fkm1
           i=k+k-3
    30     fk=fk+deltaf
           if(fk.gt.flast) goto 40
           i=i+2
           coeff=0.5d0-0.5d0*dcos(pidr*(freqn-fk))
           sample(i)=  sample(i)  *coeff
           sample(i+1)=sample(i+1)*coeff
           goto 30
                  ! recovering (weighted) time sequence by inverse FFT
    40     call FastFourierTransform (sample,a(n5),ntm1,+1)
           if(j.eq.1) then
             do i=1,ntm1
               accel(n,i)=sample(i)/ntm1
             enddo
             accel(n,nt)=accel(n,ntm1)
           else
             do i=1,ntm1
               displ(n,i)=sample(i)/ntm1
             enddo
             displ(n,nt)=displ(n,ntm1)
           endif
   100   continue
         if(iflag0) then                                   ! Saving results
           if(istyp14) then
             write (9) accel
             write (9) displ
           else
             write (9) displ
           endif
         else
           write (9) displ
         endif
   200 continue

  2001 format(///
      * ' E R R O R ! ',
      * ' NUMBER OF TIME INTERVALS (',i5,') IS LESS THAN 8',
      * ' OR NOT DIVISIBLE BY 4. SECULAR SOLUTION TERMINATED'///)
       end
```

4. Numerical illustrations

We consider effects of the secularity elimination on a solution in the problem of wave propagation in a cantilever bar subjected to a time-dependent axial loading of the Heaviside's type. Initial conditions of the equations of motion are set to be homogeneous. Deterministic input data are: cross-sectional area A = 6, length l = 1000, mass density ρ = 0.00776, the Heaviside's forcing function $Q(\tau)$ = $25.0{\cdot}10^4$. The expectation and correlation functions of the random Young modulus E_ρ along the axis of the bar are assumed as

$$E\left[E_\rho\right] = E^0\left(1 + \frac{x_\rho}{\vartheta l}\right)$$

$$\mu\left(E_\rho, E_\sigma\right) = \exp\left(-\frac{\left|x_\rho - x_\sigma\right|}{\vartheta l}\right)$$

with the correlation length ϑ=10 and the coefficient of variation α=1. The bar is modelled by 320 stochastic finite elements of equal length and 320 random variables, ρ,σ =1,2,…320, with x_ρ being ordinates of the finite element midpoints. In the analysis the first (lowest) 12 structural modes are used, representing about 98 per cent of the system total energy. The number of time steps is equal to 1024 with the sampling interval Δt=$2.8{\cdot}10^{-4}$. The elimination of secularities is performed on sequences of 1024 Fourier terms. The effect of the elimination of the secular terms on the computed expectations and variances at the free end of the bar is shown in Fig. 1.



Fig. 1. 320-element cantilever bar. Secularity elimination effects
Rys. 1. Wspornik - 320 elementów skończonych. Efekty eliminacji sekularnej

The results are obtained for the cases of undamped and damped systems, with different values of the damping factor ξ being equal to 0.0, 0.02 and 0.04 and different values of the frequency range r = 0.0, 0.17 and 0.30 for eliminating the secular terms. It is observed that only 12 highest random normal modes, which correspond to the largest 8 variances from the set of 32-uncorrelated random modes, are sufficient to approximate the random field with an error of less than 1.5 per cent. It is pointed out here that by using the Monte Carlo simulation method, with 800 randomly generated realizations required to obtain results of the same order of accuracy, it would take about 250 times the computation cost when the problem was to be solved by the direct step-by-step integration technique.

5. Concluding remarks

In general, the formal solutions to time-varying systems obtained on the basis of the Taylor expansion need not converge. However, such sequential solutions are often more effective than those based on uniformly and absolutely convergent series, which is because a power expansion may give an adequate approximation by employing only several terms. That is why they have been used extensively in many problems of applied mathematics and, particularly, in the field of computational engineering. In this context, a numerical algorithm for secularity elimination stands for an inseparable component.

6. References

[1] A.H. Nayfeh: Perturbation Methods, Wiley, New York 1973.
[2] W.K. Liu, T. Belytschko, A. Mani: Random field finite elements, Int. J. Num. Meth. Eng., 23:1831-1845, 1986.
[3] M. Kleiber, T.D. Hien: The Stochastic Finite Element Method, Wiley, New York 1992.
[4] N.M. Bogoliubov: Asymptotic Methods in the Theory of Non-Linear Oscillators, Gordon & Breach, Stanford 1961.
[5] A.V. Oppenheim, R. Schafer: Digital Signal Processing, Prentice-Hall, New Jersey 1975.
[6] L.R. Rabiner, B. Gold: Theory and Application of Digital Signal Processing, Prentice-Hall, Englewood Cliffs 1975.
[7] W.K. Liu, G.H. Besterfield, T. Belytschko: Transient probabilistic systems, Int. J. Comput. Meth. Appl. Mech. Eng., 67:27-54, 1988.
[8] T.D. Hien: A base-two fast Fourier transform algorithm, J. Pomiary Automatyka Kontrola, 413- 416, 7-2008.

Artykuł recenzowany

INFORMACJE

VII Konferencja Podstawowe Problemy Metrologii

VII Konferencja Podstawowe Problemy Metrologii (PPM'09) odbyła się w Suchej Beskidzkiej, w dniach 10–13.05.2009 roku. Organizacji Konferencji podjęło się Wydawnictwo PAK, tradycyjnie wraz z Katedrą Metrologii, Elektroniki i Automatyki Politechniki Śląskiej. Konferencja PPM miała poparcie takich instytucji jak Komitet Metrologii i Aparatury Naukowej PAN, Komisja Metrologii Oddziału PAN w Katowicach, Wojskowy Nadzór Metrologiczny, Polskie Centrum Akredytacji, Polska Sekcja IEEE oraz Polskie Stowarzyszenie Pomiarów, Automatyki i Robotyki POLSPAR.

Celem Konferencji była wymiana poglądów i rozwój współpracy między środowiskiem metrologów nauczających i metrologów praktyków, skupionych w krajowej służbie miar oraz w laboratoriach akredytowanych. Problematyka Konferencji była aktualna i wzbudziła zainteresowanie w środowisku metrologicznym, o czym świadczy szeroka akceptacja jej formuły i liczny udział uczestników poprzednich konferencji PPM jak i nowych, zainteresowanie tematyką konferencji oraz dobry poziom prac. Konferencja koncentrowała się na zagadnieniach podstawowych, wspólnych dla wszystkich rodzajów pomiarów, ale były także prezentowane aktualne zagadnienia rozwoju nowych narzędzi pomiarowych, zwłaszcza o małej niedokładności, oraz zagadnienia teorii pomiaru i narzędzi matematycznych do obróbki wyników pomiarów. Tematyka VII Konferencji PPM obejmowała zagadnienia właściwego opisu niedokładności pomiarów, zagadnienia teorii, konstrukcji i badania systemów pomiarowych, problematyki wzorców, techniki pomiarów dokładnych, przetwarzania sygnałów pomiarowych oraz bezprzewodowego przesyłu informacji. Autorami prac zebranych w materiałach konferencyjnych byli specjaliści w zakresie pomiarów różnych wielkości fizycznych i chemicznych, pracujący na unikatowych w skali kraju stanowiskach pomiarowych i posiadający duże doświadczenie praktyczne. Doskonalenie konstrukcji wzorców, metod i technik ich sprawdzania oraz rozwój układów pomiarowych o najwyższych wymaganiach metrologicznych jest ważnym obszarem współczesnej metrologii. Na rynku są przyrządy pomiarowe nowej generacji wielofunkcyjne, zautomatyzowane, zinformatyzowane, przystosowane do współpracy w sieci komputerowej, których produkcja wymaga wiedzy i umiejętności z obszaru high-tech, a jednocześnie sprzęt pomiarowy o dużych możliwościach stał się powszechnie dostępny. Także tego obszaru dotyczyły prace VII Konferencji PPM. Prace nadesłane przez autorów zaprezentowano i dyskutowano na dwunastu sesjach tematycznych. Ponadto odbyła się dyskusja okrągłego stołu na temat: „Trendy rozwojowe metrologii i kierunki zmian polskiej administracji miar", której moderatorami byli prof. dr hab. inż. Krzysztof Gniotek i płk. rez. mgr inż. Stanisław Dąbrowski. Miała ona szczególne znaczenie ze względu na toczącą się obecnie w kraju dyskusję o przyszłości polskiej metrologii i oczekiwanych zmianach w strukturach odpowiedzialnych za jej stosowanie i rozwój. W dyskusji uczestniczył m.in. wiceprezes GUM Włodzimierz Popiołek, kierownicy działów laboratoriów badawczych i wzorcujących PCA, panowie Ryszard Malesa i Tadeusz Matras, oraz szereg osób o uznanej pozycji i dorobku w dziedzinie metrologii. Komisja Konkursowa złożona z przewodniczących sesji wyróżniła cztery prezentacje, uwzględniając aktualność prezentowanej problematyki, wkład własny Autora w rozwój problematyki, formę prezentacji oraz dyskusję. Dyplomy i drobne upominki otrzymali: 1. Maryna Galowska z Technicznego Uniwersytetu w Kijowie, 2. Rafał Jarosz z Głównego Urzędu Miar w Warszawie, 3. Roman Jędrzejewski z Okręgowego Urzędu Miar w Łodzi, 4. Dr inż. Józef Kwiczala z Pol. Śląskiej i mgr inż. Aleksandra Kolano – Burian. Program Konferencji opracował 19-osobowy Komitet Programowy. W konferencji wzięło udział ponad 80 uczestników z kraju i z Ukrainy. Autorami lub współautorami prac było 75 osób. Materiały Konferencji zostały wydane przez Komisję Metrologii Oddziału Polskiej Akademii Nauk w Katowicach, w serii Konferencje, nr 14. Opublikowano 49 rozszerzonych streszczeń referatów, wszystkie które uzyskały pozytywne recenzje opracowane przez Recenzentów. Autorzy 31 referatów zostali zaproszeni do opracowania artykułów naukowych, rozszerzonych o dodatkowe wyniki i wnioski wynikające z dyskusji. Będą one opublikowane w krajowych miesięcznikach naukowych.

Wśród uczestników Konferencji panowała przyjazna atmosfera, sprzyjająca dyskusjom i swobodnej wymianie poglądów. Pierwszego dnia Konferencji do kolacji, która trwała do późnych godzin nocnych przygrywał zespół góralski, a drugiego dnia po obradach uczestnicy wysłuchali koncertu artysty Artura Thomasa, który grał na fletni Pana. Na pewno także te wrażenia artystyczne na długo pozostaną w pamięci uczestników.

Opracowanie: Tadeusz SKUBIS